

1 等号 (Equals)

1.1 基本事項

等号は、数学でいろいろな意味を持つことを中学校で学ぶ。それぞれの状況による意味の違いを人間は適当に判断できるが、プログラムである Maple では無理。Maple では、それぞれ違った記号や操作として用意され、人間が Maple に指示する必要がある。

1.1.1 変数への代入 := (colonequal)

変数に値を代入する時には := (colonequal) を使う。例えば、

a=3, b=2 のとき, a+b はいくらか?

という問題を, Maple で解かす時には,

a に 3, b に 2 を代入したとき, a+b はいくらか?

と読み直し,

```
> a:=3; # 3
> b:=2; # 2
> a+b; # 5
```

式の定義も同様。以下は $a * x \dots = c * x^2 \dots$ という式を eq1 と定義している。

```
> eq1:=a*x+b=c*x^2+d*x+e;
```

$$3x + 2 = cx^2 + dx + e$$

■**変数の初期化 (restart)** 一度何かを代入した変数を何も入れていない状態に戻す操作を変数の初期化という。すべての変数を一度に初期化するには、

```
> restart;
```

とする。なにか新たなひとまとまりの作業をするときには、このコマンドを冒頭に入れることを習慣づけるように。作業の途中でひとつの変数だけを初期化するには、シングルクォート' でくくる。

```
> a:='a';
```

a

一時的代入に subs がある。

1.1.2 方程式の解 (solve)

3x=2 を満たす x をもとめよ。

という問題は、

```
> solve(3*x=2,x);
```

$$\frac{2}{3}$$

連立方程式は以下のとおり。

```
> solve({x+y=1,x-y=2},{x,y});
```

$$\left\{x = \frac{3}{2}, y = -\frac{1}{2}\right\}$$

ただし, solve だけでは, x,y に値は代入されない.

```
> sol1:=solve({x+y=1,x-y=2},{x,y});
```

```
> assign(sol1);
```

$$sol1 := \left\{x = \frac{3}{2}, y = -\frac{1}{2}\right\}$$

とすることがある. 確認してみると

```
> x,y;
```

$$\frac{3}{2}, -\frac{1}{2}$$

となり, 値が代入されていることがわかる.

■方程式の数値解 (fsolve) 解析的に解けない場合は, 数値的に解を求める fsolve を使う. 上で x に assign しているので, x を初期化している.

```
> x:='x';
```

```
> fsolve(log(x)-exp(-x),x);
```

$$x := 1.309799586$$

1.1.3 恒等式 (Identity)

式の変形にも等号が使われる. 例えば,

$$(x-2)^2 = x^2 - 4x + 4$$

というのが等号で結ばれている. 式の変形とは, 変数 x がどんな値であっても成り立つ恒等的な変形である.

この式変形も, 問題としては,

(x-2)²を展開 (expand) せよ

と与えられるので, そのまま Maple コマンドに読み替えて

```
> expand( (x-2)^2 );
```

$$x^2 - 4x + 4$$

とすればよい. 因数分解 (factor) や微分 (diff)・積分 (int) も同様に等号で結ばれるが, Maple には操作を指示する必要がある. 詳しくは他の単元で.

1.2 練習

1.2.1 等号についての課題

1. a=3, b=4 として a,b の四則演算をおこなえ. また, べき乗 a^b を求めよ.
2. eq1=3*x+4=2*x-2, a=2 とした場合の eq1/a,eq1+a を試し, 両辺を観察せよ.
3. 3点 (1,2),(-3,4),(-1,1) を通る 2次方程式を求めよ.
4. 方程式 $\sin(x+1) - x^2 = 0$ の2つの解を fsolve のヘルプを参照して求めよ.
5. 関数 $y = \exp(x)$ の逆関数を求めよ. 2つの関数および $y=x$ を同時に $x=-5..5, y=-5..5$ で plot し, $y=x$ に対し 2つの関数が対称であることを確認せよ.

1.3 解答例

1.3.1 Equals

1. $a=3$, $b=4$ として a, b の四則演算をおこなえ. また, べき乗 a^b を求めよ.

```
> a:=3:
> b:=4:
> a+b;a-b;a*b;a/b;a^b;      #省略
```

2. $eq1=3*x+4=2*x-2$, $a=2$ とした場合の $eq1/a, eq1+a$ を試し, 両辺を観察せよ.

```
> eq1:=3*x+4=2*x-2;
> a:=2;
> eq1/a;
> eq1+2;
```

$$eq1 := 3x + 4 = 2x - 2$$

$$a := 2$$

$$\frac{3}{2}x + 2 = x - 1$$

$$3x + 6 = 2x$$

(1)

3. 3 点 $(1,2), (-3,4), (-1,1)$ を通る 2 次方程式を求めよ.

まず 2 次関数を定義する.

```
> restart;
> f:=x->a*x^2+b*x+c;
```

$$f := x \mapsto ax^2 + bx + c$$

$(1,2)$ を通ることから, $f(1)=2$ が成立. これを $eq1$ として保存.

```
> eq1:=f(1)=2;
```

$$eq1 := a + b + c = 2$$

他の点も同様に

```
> eq2:=f(-3)=4;
> eq3:=f(-1)=1;
```

$$eq2 := 9a - 3b + c = 4$$

$$eq3 := a - b + c = 1$$


この 3 個の連立方程式から, a, b, c を求めれば解となる.

```
> solve({eq1,eq2,eq3},{a,b,c});
```

$$\{a = 1/2, b = 1/2, c = 1\}$$

4. 方程式 $\sin(x+1) - x^2 = 0$ の2つの解を fsolve のヘルプを参照して求めよ.
 まず, 2つの関数とみなしてプロット.

```
> plot([sin(x+1),x^2],x=-1..1);
```



解が2つあることに注意. 与えられた関数値が0となる方程式として定義し, これを solve でとく.

```
> eq1:=sin(x+1)-x^2=0;
> solve(eq1,x);
```

$$eq1 := \sin(x+1) - x^2 = 0$$

$$-1 + \text{RootOf}(-\sin(Z) + 1 - 2Z + Z^2)$$

これでは解を求めてくれないので, fsolve で数値解を求める.

```
> fsolve(eq1,x);
```

0.9615690350

これでは x の負にあるもう一つの解がでない. これを解決するには, fsolve で x に初期値を入れて実行する.

```
> fsolve(eq1,x=-1..0);
```

-0.6137631294

5. 関数 $y = \exp(x)$ の逆関数を求めよ. 2つの関数および $y=x$ を同時に $x=-5..5, y=-5..5$ で plot し, $y=x$ に対して2つの関数が対称であることを確認せよ.
 先ず, 与関数を f(x) として定義しておく.

```
> f:=x->exp(x);
```

$$f := x \mapsto e^x$$

$y=f(x)$ として, x について解いてみる.

```
> eq1:=y=f(x);
> solve(eq1,x);
```

$$eq1 := y = e^x$$

$$\ln(y)$$

うまく y の関数として解けている. これを eq2 と定義し直して, 逆関数として定義する.


```
> eq2:=solve(eq1,x);
> invf:=unapply(eq2,y);
```

$$eq2 := \ln(y)$$

$$invf := y \mapsto \ln(y)$$

同時にプロットしてみる.

```
> plot([f(x), invf(x), x], x=-5..5, y=-5..5);
```



$y=x$ を軸として, 2 つの関数が対称であることが確認できる.