

## 0.1 初等関数とそのほかの関数 (Functions)

[[[解説]]]

### 0.1.1 初等関数 (Elementary Functions)

**四則演算と evalf** 四則演算は”+-\*/”. 割り切れない割り算は分数のまま表示される.

```
> 3/4;
```

$$\frac{3}{4}$$

強制的に数値 (浮動小数点数) で出力するには evalf を用いる.

```
> evalf(3/4);
```

$$0.7500000000$$

**多項式関数 (polynom)** かけ算も省略せずに打ち込む必要がある. またべき乗は"^"である.

```
> 3*x^2-4*x+3;
```

$$3x^2 - 4x + 3$$

**平方根 (sqrt)** 平方根は square root を略した sqrt を使う.

```
> sqrt(2);
```

$$\sqrt{2}$$

**三角関数 (trigonal)** sin, cos などの三角関数はラジアンで入力する. ただし,  $\sin^2 x$  などは

```
> sin^2 x;
```

```
Error, missing operator or ‘;’
```

ではだめで,

```
> sin(x)^2;
```

$$\sin^2 x$$

と省略せずに打ち込まねばならない. 三角関数でよく使う定数  $\pi$  は”Pi”と入力する. Maple は大文字と小文字を区別するので注意.

ラジアン (radian) に度 (degree) から変換するには以下のようにする.

```
> convert(90*degrees, radians);
```

```
convert(1/6*Pi, degrees);
```

$$\frac{1}{2}\pi$$

$$30 \text{ degrees}$$

**その他の関数 (inifnc)** その他の初等関数やよく使われる超越関数など, Maple の起動時に用意されている関数のリストは,

```
> ?inifnc;
```

で得られる.

## ユーザー定義関数 (unapply)

初等関数やその他の関数を組み合わせてユーザー定義関数を作ることができる.


関数  $f(x) = 2x - 3$  とおくとする場合, Maple では,

```
> f:=x->2*x-3;
```

$$f := x \rightarrow 2x - 3$$

と, 矢印で書く. これが関数としてちゃんと定義されているかは, いくつかの数値や変数を  $f(x)$  に代入して確認する.

```
> f(3);           #3
f(a);             #2 a - 3
plot(f(x),x=-2..2);
```



もう一つ関数定義のコマンドとして次の unapply も同じ意味である.

```
> f:=unapply(2*x-3,x);
```

$$f := x \rightarrow 2x - 3$$

ただし, 矢印での定義ではときどき変な振る舞いになるので, unapply を常に使うようにこころがけたほうが安全.

## package の呼び出し (with)

Maple が提供する膨大な数の関数から, 目的とするものを探し出すには help を使う. 普段は使わない関数は, 使う前に明示的に呼び出す必要がある. 例えば, 線形代数によく使われる関数群は,

```
> with(LinearAlgebra):
```

としておく必要がある. この他にもいくつかの有益な関数パッケージが用意されている.

```
> ?index[package];
```

で用意されているすべての package が表示される.

[[課題]]

## 関数についての課題

1. evalf のヘルプを参照して, Pi を 1000 桁まで表示せよ.
2. 正接関数 (tan) とその逆関数 x/tan を  $x=-\text{Pi}/2..\text{Pi}/2, y=-\text{Pi}..\text{Pi}, \text{scaling}=\text{constrained}$  で同時にプロットせよ.
3. 対数関数は  $\ln(x)$  で与えられる. ヘルプを参照しながら次の値を求めよ.

$$\log_{10} 1000, \log_2 \frac{1}{16}, \log_{\sqrt{5}} 125$$

4. 次の関数は  $y = 2^x$  とどのような位置関係にあるか  $x=-5..5, y=-5..5$  で同時にプロットして観察せよ.

$$y = -2^x, y = (1/2)^x, y = -(1/2)^x$$

5. 指数関数は exp で与えられる.  $e^x$  と  $\log x$  関数および  $y = x$  を同時に  $x=-5..5, y=-5..5$  で plot せよ. またそれらの関数の位置関係を述べよ.
6. 階乗関数 factorial に 3 を代入して何を求める関数か予測せよ. ヘルプを参照し, よりなじみの深い表記を試してみよ.

## [[[解答例]]]


### Functions

1. evalf のヘルプを参照して, Pi を 1000 桁まで表示せよ.

```
> ?evalf;  
> evalf[1000](Pi);      #省略
```

2. 正接関数 (tan) とその逆関数 x/tan を  $x=-\text{Pi}/2..\text{Pi}/2, y=-\text{Pi}..\text{Pi}, \text{scaling}=\text{constrained}$  で同時にプロットせよ.

```
> plot([tan(x), arctan(x)], x=-Pi/2..Pi/2, y=-Pi..Pi, scaling=constrained);
```



3. 対数関数は  $\ln(x)$  で与えられる. ヘルプを参照しながら次の値を求めよ.

$$\log_{10} 1000, \log_2 \frac{1}{16}, \log_{\sqrt{5}} 125$$

```
> ?ln;  
  
> log10(1000);
```

```
> log[2](1/16);
```

−4


```
> log[sqrt(5)](125);
```

4. 次の関数は  $y = 2^x$  とどのような位置関係にあるか  $x=-5..5, y=-5..5$  で同時にプロットして観察せよ.

$$y = -2^x, y = (1/2)^x, y = -(1/2)^x$$


```
> plot([2^x, -2^x, (1/2)^x, -(1/2)^x], x=-5..5, y=-5..5);
```

注目している関数以外を消せばはっきりするが, i)  $x$  軸に対称, ii)  $y$  軸に対称, iii) 原点に対称.



5. 指数関数は  $\exp$  で与えられる.  $e^x$  と  $\log x$  関数および  $y = x$  を同時に  $x=-5..5, y=-5..5$  で plot せよ. またそれらの関数の位置関係を述べよ.

```
> plot([exp(x), log(x), x], x=-5..5, y=-5..5);
```



6. 階乗関数 factorial に 3 を代入して何を求める関数か予測せよ. ヘルプを参照し, よりなじみの深い表記を試してみよ.

```
> factorial(3)
```

```
> ?factorial;
```

```
> 3!;
```