第1章　微積分

1.1　微分 (Diff)-I

[[[解説]]]

1.1.1　単純な微分 (diff)

単純な一変数関数の一次微分は，以下の通り．

```
> diff(x^2-3*x+2,x); #res: 2x-3
```

高次の微分は，微分変数を必要なだけ並べる．

```
> diff(sin(x),x,x); #res: -sin(x)
```

さらに高次では次のように$を使った記法が便利．これは x についての 3 次微分を表わす．

```
> diff(x^4,x$3); #res: 24x
```

偏微分 (PartialDiff)　複数の変数を持つ多変数の関数では，微分する変数を明示すれば偏微分が求められる．

```
> eq1:=(x+y)/(x*y);
> diff(eq1,x);
```

$$eq1 := \frac{x+y}{xy}$$

$$\frac{1}{xy} - \frac{x+y}{x^2 y}$$

[[[例題]]]

1.1.2　例題:関数の微分と増減表

次の関数とその 1 次導関数を同時にプロットし概形を確認し，さらに増減表を求めよ．

$$\frac{x}{x^2-2x+4}$$

解答例

```
> f0:=unapply(x/(x^2-2*x+4),x):
> df:=unapply(diff(f0(x),x),x);
> plot([f0(x),df(x)],x);
```

$$df := x \mapsto \left(x^2-2\,x+4\right)^{-1} - \frac{x\,(2\,x-2)}{\left(x^2-2\,x+4\right)^2} \tag{1.1}$$



1.1.3　例題:接線 (Tangent)

次の関数の $x = 3$ での接線を求め，2 つの関数を同時にプロットせよ．

$$y = x^3 - 2\,x^2 - 35\,x$$

解答例　与関数を f0 と定義.

```
> f0:=unapply(x^3 - 2*x^2 - 35*x,x);
```

$$f0 := x \mapsto x^3 - 2\,x^2 - 35\,x$$

微分関数を df と定義

```
> df:=unapply(diff(f0(x),x),x);
```

$$df := x \mapsto 3\,x^2 - 4\,x - 35$$

接点 (x0,f0(x0)) で傾き df(x0) の直線を f1 と定義.

```
> x0:=3;
> eq1:=df(x0)*(x-x0)+f0(x0);
> f1:=unapply(eq1,x);
```

$$x0 := 3$$

$$eq1 := -20\,x - 36$$

$$f1 := x \mapsto -20\,x - 36$$

2 つの関数を同時にプロット.

```
> plot([f0(x),f1(x)],x=-5..5);
```



1.2 微分 (Diff)-II

[[[解説]]]

1.2.1 級数展開 (series)

Taylor 級数は以下のようにして，中心点 (x=a)，次数 (4 次) を指定する．

```
> t1:=series(sin(x),x=a,4);
```

$$t1 := \sin(a) + \cos(a)(x-a) - \frac{1}{2}\sin(a)(x-a)^2 - \frac{1}{6}\cos(a)(x-a)^3 + O((x-a)^4)$$

誤差の次数を示す $O((x-a)^4)$ があるため，このままでは関数として使えない．unapply で定義関数として使うためには，convert を使って多項式 (polynomial) に変換しておく．

```
> e1:=convert(t1,polynom);
> f1:=unapply(e1,x);
```

$$t1 := \sin(a) + \cos(a)(x-a) - \frac{1}{2}\sin(a)(x-a)^2 - \frac{1}{6}\cos(a)(x-a)^3$$

$$f1 := x \mapsto \sin(a) + \cos(a)(x-a) - \frac{1}{2}\sin(a)(x-a)^2 - \frac{1}{6}\cos(a)(x-a)^3$$

1.2.2 全微分 (D)

全微分を計算するときは，D を用いる．

```
> f:=unapply(x^4*exp(-y^2),(x,y));
> D(f(x,y));
> (D@@2)(f(x,y));
```

$$f := (x,y) \mapsto x^4 \exp(-y^2)$$

$$4\,D(x)\,x^3 \exp(-y^2) + x^4 D(\exp(-y^2))$$

$$4\left(D^{(2)}\right)(x)\,x^3 \exp(-y^2) + 12\,(D(x))^2\,x^2 \exp(-y^2) + 8\,D(x)\,x^3 D(\exp(-y^2)) + x^4 \left(D^{(2)}\right)(\exp(-y^2))$$

ここで，D(x) などは x の全微分を表わす．これは，x,y を変数としているので

```
> diff(x,x);
> diff(exp(-y^2),y);
```

$$1$$

$$-2\,y \exp(-y^2)$$

であるが Maple には分からない．そこで全微分の最終形を得るには，あらかじめ D(x) などの結果を求めておき，subs で明示的に代入する必要がある．

```
> dd:=D(f(x,y)):
> eqs:={D(x)=diff(x,x),D(exp(-y^2))=diff(exp(-y^2),y)};
> subs(eqs,dd);
```

$$eqs := \{D(x) = 1, D(\exp(-y^2)) = -2\,y \exp(-y^2)\}$$

$$4\,x^3 \exp(-y^2) - 2\,x^4 y \exp(-y^2)$$

1.2.3　複合関数の微分

```
> diff(f(x)*g(x),x);
> diff(f(g(x)),x);
```

$$\left(\frac{d}{dx} f(x)\right) g(x) + f(x) \frac{d}{dx} g(x)$$

$$\mathrm{D}(f)(g(x)) \frac{d}{dx} g(x)$$

```
> f:=x->exp(x);
> g:=x->cos(x);
> diff(f(x)*g(x),x);
> diff(f(g(x)),x);
```

$$f := x \mapsto \exp(x)$$

$$g := x \mapsto \cos(x)$$

$$\exp(x)\cos(x) - \exp(x)\sin(x)$$

$$-\sin(x)\exp(\cos x)$$

[[[課題]]]

微分に関する課題

1. 次の関数を微分せよ．

   i)$x \log x$, ii) $\dfrac{1}{(1+x)^3}$, iii) $\sqrt{4x+3}$, iv) $\dfrac{1}{a^2+(x-x_0)^2}$
2. 次の関数の 1 次から 5 次導関数を求めよ．

   i) $\sin^2 x$, ii) e^x
3. 以下の関数を x0 まわりで 3 次までテイラー展開し，得られた関数ともとの関数をプロットせよ．さらに 5 次まで展開した場合はどう変化するか．

   i) $y = \sin x, x_0 = 0$, ii) $y = \cos x, x_0 = \dfrac{\pi}{2}$
4. (発展課題) $f(x, y) = e^x log(1+y)$ を $x = 0$, $y = 0$ のまわりで 3 次まで展開せよ．

[[[解答例]]]

Diff

1.
```
> diff(x*log(x),x);
> diff(1/(1 + x)^3,x);
> diff(sqrt(4*x + 3),x);
>diff(1/(a^2+(x-x0)^2),x);
```

$$\ln(x) + 1$$

$$-3\,(1+x)^{-4}$$

$$2\,\left(\sqrt{4x+3}\right)^{-1}$$

$$-\frac{2x - 2x0}{\left(a^2 + (x - x0)^2\right)^2}$$

2.
```
> diff(sin(x)^2,x);
> diff(sin(x)^2,x$2);
```

$$2\ \sin(x)\cos(x)$$
$$2\ (\cos(x))^2 - 2\ (\sin(x))^2$$

以下略

3. 先ず与関数を f0 と定義

```
> f0:=unapply(sin(x),x);
```

$$f0\ := x \mapsto \sin(x)$$

テイラー展開した結果を eq1 とする．関数として定義するために eq1 を多項式に変換し (convert)，unapply をかける．

```
> eq1:=series(f0(x),x=0,3);
> f1:=unapply(convert(eq1,polynom),x);
```

$$eq1\ := x + O\left(x^3\right)$$
$$f1\ := x \mapsto x$$

５次についても同様

```
> eq2:=series(f0(x),x=0,5);
> f2:=unapply(convert(eq2,polynom),x);
```

$$eq2\ := x - 1/6\,x^3 + O\left(x^5\right)$$
$$f2\ := x \mapsto x - 1/6\,x^3$$

3 つの関数を同時プロット

```
> plot([f0(x),f1(x),f2(x)],x=-Pi..Pi);
```




```
> series(f0(x),x=Pi/2,3)
```

以外は前問とおなじ．

4. `f:=unapply(exp(x)*log(1+y),(x,y));`

$$f := (x, y) \mapsto \exp(x)\ln(1+y)$$

```
eq1:=series(series(f(x,y),x=0,3),y=0,3);
```

$$eq1 := O\left(x^3\right) + \left(1 + 1/2\,x^2 + x\right)y + \left(-1/2\,x - 1/2 - 1/4\,x^2\right)y^2 + O\left(y^3\right)$$

```
> g:=unapply(convert(convert(eq1,polynom),polynom),(x,y));
```

$$g := (x, y) \mapsto \left(1 + 1/2\,x^2 + x\right)y + \left(-1/2\,x - 1/2 - 1/4\,x^2\right)y^2$$

```
> plot3d([f(x,y),g(x,y)],x=-1..1,y=-1..1,axes=box);
```



きれいな表示

以下のようにすると表示がきれい.

```
> f:=unapply(x^4*exp(-y^2),(x,y));
> d:=Diff(f(x,y),x);
> d=value(d);
```

$$f := (x, y) \mapsto x^4 \exp(-y^2)$$

$$d := \frac{\partial}{\partial x}\left(x^4 \exp(-y^2)\right)$$

$$\frac{\partial}{\partial x}\left(x^4 \exp(-y^2)\right) = 4\,x^3 \exp(-y^2)$$