

# 第1章 式変形

## 1.1 数式処理コマンドの分類 (EquationManipulationTable)

[[[解説]]]

数式の変形は、手で直すほうが圧倒的に早くきれいになる場合が多い。しかし、テイラー展開や、複雑な積分公式、三角関数と exp 関数の変換などの手間がかかるところを、Maple は間違いなく変形してくれる。ここで示すコマンドを全て覚える必要は全くない。というか忘れるもの。ここでは、できるだけコンパクトにまとめて、悩んだときに参照できるようにする。初めての人は、ざっと眺めた後、鉄則からじっくりフォローせよ。

表 1.1: 数式処理で頻繁に使うコマンド.

式の変形	式の分割抽出	省略操作, その他	代入, 置換, 仮定
simplify:簡単化	lhs, rhs:左辺, 右辺	:連結作用素	subs:一時的代入
expand:展開	numer, denom:分子, 分母	seq:for-loop の簡易表記	restart,a:='a':初期化
factor:因数分解	coeff:係数	map:関数の要素への適用	assume:仮定
normal:約分・通分	nops, op	add,mul:単純な和, 積	assuming:一時的仮定
combine:公式でまとめる		sum,product:数式に対応した和, 積	assign:値の確定
collect:次数でまとめる		limit:極限	about:仮定の中身
sort:昇べき, 降べき			anames('user'):使用変数名
rationalize:分母の有理化			
convert:形式の変換			

このほかにも, solve(解), diff(微分), int(積分), series(級数展開) 等は頻繁に数式の導出・変形に登場する.

## 1.2 式の変形に関連したコマンド (Simplify)

[[解説]]

### 1.2.1 簡単化 (simplify)

```
> simplify(exp1, 副関係式):  
> simplify(3*x+4*x+2*y);
```

$$7x + 2y$$

```
> exp1:=3*sin(x)^3-sin(x)*cos(x)^2;  
> simplify(exp1);
```

$$\begin{aligned} \text{exp1} &:= 3 (\sin(x))^3 - \sin(x) (\cos(x))^2 \\ &\quad - \left( 4 (\cos(x))^2 - 3 \right) \sin(x) \end{aligned}$$

```
> simplify(exp1, {cos(x)^2=1-sin(x)^2});
```

$$4 (\sin(x))^3 - \sin(x)$$

オプションとして size を指定するとより簡単になる場合がある.

```
> simplify(exp1, size):
```

### 1.2.2 展開 (expand)

```
> expand((x+y)^2);
```

$$x^2 + 2xy + y^2$$

### 1.2.3 因数分解 (factor)

```
> factor(4*x^2-6*x*y+2*y^2);
```

$$2 (2x - y) (x - y)$$

#### 1.2.4 約分・通分 (normal)

```
> normal((x+y)/(x^2-3*x*y-4*y^2));
```

$$\frac{1}{x-4y}$$

```
> normal(1/x+1/y);
```

$$\frac{y+x}{xy}$$

#### 1.2.5 項を変数でまとめる (collect)

```
> collect(4*a*x^2-3*y^2/x+6*b*x*y+3*c*y+2*y^2,y);
```

$$(-3x^{-1}+2)y^2+(6bx+3c)y+4ax^2$$

#### 1.2.6 項を公式でまとめる (combine)

```
> combine(sin(x)^2+3*cos(x)^2);
```

$$2+\cos(2x)$$

#### 1.2.7 昇べき, 降べき (sort)

```
> sort(exp1,[x,y]);
```

```
> sort(exp1,[x],opts);
```

opts=tdeg(総次数順), plex(辞書式順), ascending(昇順), descending(降順)

```
> exp1:=x^3+4*x-3*x^2+1;
```

```
> sort(exp1);
```

$$x^3-3x^2+4x+1$$

```
> sort(exp1,[x],ascending);
```

$$1+4x-3x^2+x^3$$

```
> exp2:=x^3-a*x*y+4*x^2+y^2:
> sort(exp2);
```

$$-axy + x^3 + 4x^2 + y^2$$

```
> sort(exp2,[x]);
```

$$x^3 + 4x^2 - ayx + y^2$$

```
> sort(exp2,[a,y]);
> sort(exp2,[a],plex);
```

$$\begin{aligned} & -xay + y^2 + x^3 + 4x^2 \\ & -xya + x^3 + 4x^2 + y^2 \end{aligned}$$

### 1.2.8 分母の有理化 (rationalize)

```
> eq1:=(1+sqrt(2))/(1-sqrt(3));
```

分母の有理化は、この式を

$$\frac{1 + \sqrt{2}}{1 - \sqrt{3}} \frac{1 + \sqrt{3}}{1 + \sqrt{3}}$$

とする事に対応します.

```
> rationalize(eq1);
```

$$-1/2 \left(1 + \sqrt{2}\right) \left(1 + \sqrt{3}\right)$$

### [[課題]]

1. 以下の式を簡単化せよ.

i)  $x^{100} - 1$ , ii)  $x^2 - y^2 + 2x + 1$ , iii)  $(a + b + c)^3 - (a^3 + b^3 + c^3)$

### [[解答例]]

```
> factor(x^100-1);
```

$$\begin{aligned} & (x - 1) (1 + x^4 + x^3 + x^2 + x) (1 + x^{20} + x^{15} + x^{10} + x^5) \\ & (1 + x) (1 - x + x^2 - x^3 + x^4) (1 - x^5 + x^{10} - x^{15} + x^{20}) \\ & (1 + x^2) (x^8 - x^6 + x^4 - x^2 + 1) (x^{40} - x^{30} + x^{20} - x^{10} + 1) \end{aligned}$$

```
> factor( x^2-y^2+2*x+1);
```

$$(x + 1 + y) (x + 1 - y)$$

```
> factor((a+b+c)^3-(a^3+b^3+c^3));
```

$$3 (b + c) (c + a) (a + b)$$

## 1.3 式の分割抽出 (convert)

[[[解説]]]

### 1.3.1 形式の変換 (convert(exp1,opt))

数式の記述形式を変えるのに頻繁に使う.

opt	意味
polynom	級数を多項式 (polynomial) に
trig	三角関数 (trigonal) に
sincos	tan を含まない, sin,cos に
exp	指数関数形式に
parfrac	部分分数 (partial fraction) に
rational	浮動小数点数を有理数形式に

```
> s1:=series(sin(x),x,4);  
> convert(s1,polynom);
```

$$s1 := x - \frac{1}{6}x^3 + O(x^4)$$
$$x - \frac{1}{6}x^3$$

```
> convert(sin(x),exp);
```

$$-\frac{1}{2}I(\exp(Ix) - \exp(-Ix))$$

```
> convert(sinh(x),exp);
```

$$\frac{1}{2}\exp(x) - \frac{1}{2}\exp(-x)$$

```
> convert(tan(x),sincos);
```

$$\frac{\sin(x)}{\cos(x)}$$

```
> convert(exp(I*x),trig);
```

$$\cos(x) + I\sin(x)$$

```
> convert(1/(x-1)/(x+3),parfrac);
```

$$-\frac{1}{4(x+3)} + \frac{1}{4(x-1)}$$

```
> convert(3.14,rational);
```

$$\frac{157}{50}$$

### 1.3.2 左辺, 右辺 (lhs, rhs)

それぞれ, 左辺, 右辺を意味する left hand side, right hand side の略

```
> lhs(sin(x)^2-1-1/x);  
> rhs(sin(x)^2-1-1/x);
```

$$\sin(x)^2 - 1 - \frac{1}{x}$$

### 1.3.3 分母, 分子 (denom, numer)

それぞれ, 分母, 分子を意味する denominator, numerator の略

```
> numer(a*x/(x+y)^3);  
> denom(a*x/(x+y)^3);
```

$$\frac{ax}{(x+y)^3}$$

### 1.3.4 係数 (coeff)

```
> coeff(4*a*x^2-3*y^2/x+6*b*x*y+3*c*y+2*y^2,y^2);
```

$$-\frac{3}{x} + 2$$

### 1.3.5 要素の取りだし, 要素数 (op, nops)

op, nops は list 配列から要素や要素数を取り出すのに頻繁に使われる. 数式を含めた, より一般的な構造に対しても作用させることができる.

```
> op(4*a*x^2-3*y^2/x+6*b*x*y+3*c*y+2*y^2);
```

$$4ax^2, -\frac{3y^2}{x}, 6bxy, 3cy, 2y^2$$

```
> nops(4*a*x^2-3*y^2/x+6*b*x*y+3*c*y+2*y^2); #res: 5
```

## [[課題]]

1. 以下の関数を  $x_0$  まわりで 3 次までテイラー展開し, 得られた関数ともとの関数をプロットせよ. さらに高次まで展開した場合はどう変化するか.

i)  $y = \cos(x), x_0 = 0$ , ii)  $y = \ln(x), x_0 = 1$ , iii)  $y = \exp(-x), x_0 = 0$

2.  $\frac{x+1}{(x-1)(x^2+1)^2}$  を部分分数に展開せよ.
3.  $\frac{1}{1-x^4} = \frac{a}{x^2+1} + \frac{b}{x+1} + \frac{c}{x-1}$  が常に成立する  $a, b, c$  を定めよ.
4.  $\frac{8}{3-\sqrt{5}} - \frac{2}{2+\sqrt{5}}$  を簡単化せよ.
5.  $x^2 + 2kx + 5 - k = 0$  が重根をもつように  $k$  を定めよ.

## [[ 解答例 ]]


1. series で Taylor 展開した後, convert で多項式 (polynom) に変換する.

```
> convert(series(cos(x),x),polynom);
```

$$1 - \frac{1}{2}x^2 + \frac{1}{24}x^4$$


```
> plot([convert(series(cos(x),x),polynom),cos(x)],x=-Pi..Pi);
```

$$1 - \frac{1}{2}x^2 + \frac{1}{24}x^4 - \frac{1}{720}x^6 + \frac{1}{40320}x^8$$



高次展開する場合には, series の最後の引数に次数を指定する. デフォルトは 6 次. 関数の一致具合が向上していることに注目.

```
convert(series(cos(x),x,9),polynom);
plot([convert(series(cos(x),x,9),polynom),cos(x)],x=-Pi..Pi);
```




series での展開で, x=1 とするとその周りでの展開になる.

```
> convert(series(ln(x),x=1),polynom);
```

$$x - 1 - \frac{1}{2}(x-1)^2 + \frac{1}{3}(x-1)^3 - \frac{1}{4}(x-1)^4 + \frac{1}{5}(x-1)^5$$


```
> plot([convert(series(ln(x),x=1),polynom),ln(x)],x=0..2);
```



```
> convert(series(exp(-x),x),polynom);
```

$$1 - x + \frac{1}{2}x^2 - \frac{1}{6}x^3 + \frac{1}{24}x^4 - \frac{1}{120}x^5$$

```
> plot([convert(series(exp(-x),x),polynom),exp(-x)],x=0..2);
```



```
2. > restart; eq1:=(x+1)/((x-1)*(x^2+1)^2);
```

$$eq1 := \frac{x+1}{(x-1)(x^2+1)^2}$$

```
> convert(eq1,parfrac);
```

$$\frac{1}{2(x-1)} - \frac{x}{(x^2+1)^2} + \frac{1-x-1}{2(x^2+1)}$$

```
3. > convert(1/(1-x^4),parfrac);
```

$$-\frac{1}{4(x-1)} + \frac{1}{4(x+1)} + \frac{1}{2(x^2+1)}$$

```
4. > eq2:=8/(3-sqrt(5))-2/(2+sqrt(5));
```

$$eq2 := \frac{8}{3-\sqrt{5}} - \frac{2}{2+\sqrt{5}}$$

まずはデフォルトの簡単化 (simplify).

```
> simplify(eq2); #res: 10
```

5. 方程式をたてる.

```
> eq3:=x^2+2*k*x+(5-k);
```

$$eq3 := x^2 + 2kx + 5 - k$$

単純に解を求めて,



```
> sol1:=solve(eq3=0,x);
```

$$sol1 := -k + \sqrt{k^2 - 5 + k}, -k - \sqrt{k^2 - 5 + k}$$

それが一致する場合の k を解く.

```
> solve(sol1[1]=sol1[2],k);
```

$$-\frac{1}{2} + \frac{1}{2}\sqrt{21}, -\frac{1}{2} - \frac{1}{2}\sqrt{21}$$

別解. まず, 係数を coeff で取り出す.

```
> aa:=coeff(eq3,x^2); #res:1
```

```
> bb:=coeff(eq3,x); #res: 2 k
```

```
> cc:=coeff(eq3,x,0); #res: 5 - k
```

判別式  $D = b^2 - 4ac$  を計算して,  $= 0$  において k について解く.

```
> solve(bb^2-4*aa*cc=0,k);
```

$$-\frac{1}{2} + \frac{1}{2}\sqrt{21}, -\frac{1}{2} - \frac{1}{2}\sqrt{21}$$

## 1.4 代入, 置換, 仮定に関連したコマンド (assume, subs)

[[解説]]

### 1.4.1 一時的代入 (subs)

関係式 (x=2) を式 (exp1) に一時的に代入した結果を表示.

```
> exp1:=x^2-4*x*y+4; subs(x=2,exp1);
```

$$\begin{aligned} \text{exp1} &:= x^2 - 4xy + 4 \\ &8 - 8y \end{aligned}$$

```
> subs({x=a+2,y=sin(x)},exp1);
```

$$(a+2)^2 - 4(a+2)\sin(x) + 4$$

### 1.4.2 仮定 (assume)

変数になにかの条件を仮定するときを使う. たとえば, 根を開くときに, 与式が正と仮定すると開かれる.

```
> sqrt(b^2);  
> assume(a>0); sqrt(a^2);
```

$$\begin{aligned} &\sqrt{b^2} \\ &a \sim \end{aligned}$$

### 1.4.3 一時的仮定 (assuming)

assume と同じだが, 一時的に仮定するときに使われる.

```
> exp1:=x^2-4*x+4;  
> sqrt(exp1);
```

$$\begin{aligned} \text{exp1} &:= x^2 - 4x + 4 \\ &\sqrt{(-2+x)^2} \end{aligned}$$

```
> sqrt(exp1) assuming x>2;
```

$$-2 + x$$

assume に加えての仮定に additionally がある.

#### 1.4.4 solve で求めた値の確定 (assign)

解を求める solve をして

```
> x:='x':y:='y':  
> s1:=solve({x-y+1=0,x+y-2=0},{x,y});
```

$$s1 := \left\{ x = \frac{1}{2}, y = \frac{3}{2} \right\}$$

このまま, x,y の中身を見ても,

```
> x,y;
```

$x, y$

代入されていない. s1 を assign(確定) すると,

```
> assign(s1);  
> x,y;
```

$\frac{1}{2}, \frac{3}{2}$

と代入される. 一時的代入 subs と使い分ける.

#### 1.4.5 assume で仮定した内容の確認 (about)

```
> about(a);
```

Originally a, renamed a~:

is assumed to be: RealRange(Open(0),infinity)

#### 1.4.6 ユーザが定義した変数の確認 (anames('user'))

```
> anames('user');
```

$s1, y, x, a$

#### 1.4.7 値の初期化 (restart,a:='a')

### 1.4.8 連結作用素 (||)

前後の変数をくっつけて新たな変数とする.

```
> a||1; #res: a1  
> a||b; #res: ab
```

プログラムの中で使うとより便利.

```
> for i from 1 to 3 do  
  a||i:=i^2;  
end do;
```

$a1 := 1$

$a2 := 4$

$a3 := 9$

### 1.4.9 for-loop の簡略表記 (seq)

数列を意味する sequence の略.

```
> seq(k,k=4..7);
```

$4, 5, 6, 7$

### 1.4.10 リスト要素への関数の一括適用 (map)

```
> f:=x->exp(-x);  
> map(f,[0,1,2,3]);
```

$f := x \mapsto \exp(-x)$

$[1, \exp(-1), \exp(-2), \exp(-3)]$

上記の3つを組み合わせると, 効率的に式を扱うことができる.

```
> map(sin,[seq(theta||i,i=0..3)]);
```

$[\sin(\theta 0), \sin(\theta 1), \sin(\theta 2), \sin(\theta 3)]$

### 1.4.11 単純な和, 積 (add,mul)

```
> add(x^i,i=0..3);
```

$1 + x + x^2 + x^3$

```
> mul(x^i,i=0..3);
```

$x^6$

### 1.4.12 数式にも対応した和, 積 (sum,product)

```
> add(x^i,i=0..n);
```

```
Error, unable to execute add
```

```
> sum(x^i,i=0..n);
```

$$\frac{x^{n+1}}{x-1} - \frac{1}{x-1}$$

```
> product(x^i,i=0..n);
```

$$x^{\frac{1}{2}(n+1)^2 - \frac{1}{2}n - \frac{1}{2}}$$

### 1.4.13 極限 (limit)

```
> limit(exp(-x),x=infinity); #res: 0
```

```
> limit(tan(x),x=Pi/2);
```

*undefined*

```
> limit(tan(x),x=Pi/2,left);
```

```
> limit(tan(x),x=Pi/2,complex);
```

$$\infty$$
$$-\infty + \infty I$$

## 1.5 式の変形の基本 (BottomLine)

### [[解説]]

どうしても解かなければならない課題を前にコマンドリファレンスのあちこちを参照しながら解いていくのが数式処理を修得する最速法である。とびかかる前にちょっとした共通のコツがある。それをここでは示す。数式処理ソフトでの数式処理とは、数式処理ソフトが『自動的にやって』くれるのではなく、実際に紙と鉛筆で解いていく手順を数式処理ソフトに『やらせる』ことであることを肝に銘じよ。

### 1.5.1 鉄則

Mapleをはじめとする数式処理ソフトの習得にあたって初心者がつまづく共通の過ちを回避する鉄則がある。

**鉄則0：restart をかける** 続けて入力すると前の入力が生きている。違う問題へ移るときや、もう一度入力をし直すときには、restart; を入力して初期状態からはじめる。入力した順番が狂っている場合もある。頭から順に return をやり直す。

**鉄則1：出力してみる** 多くのテキストではページ数の関係で出力を抑止しているが、初心者が問題を解いていく段階ではデータやグラフをできるだけ多く出力する。最後のコロンをセミコロンに変える、あるいは途中に print 文を入れる。

**鉄則2：関数に値を代入してみる** 数値が返ってくるべき時に変数があればどこかで入力をミスっている。plot で以下のようなエラーが出た場合にチェック。

```
> plot(f(x),x);
```

```
Warning, unable to evaluate the function to numeric values in the region; see  
the plotting command's help page to ensure the calling sequence is correct
```

**鉄則3：内側から順に入力する** 長い入力や for-loop を頭から打ち込んではいけない!! 内側から順に何をしているか解説・確認しながら打ち込む。括弧が合わなかったり、読み飛ばしていたりというエラーが回避できる。

### 1.5.2 具体例：無限積分

以下に示す積分を実行せよ。

$$\int_{-\infty}^{\infty} x \exp(-\beta c x^2) (1 + \beta g x^3) dx$$

最新版の Maple では改良が施されていて、このような複雑な積分も一発で求まる。

```
> f1:=unapply(x*exp(-beta*c*x^2)*(1+beta*g*x^3),x);
```

$$f1 := x \mapsto x \exp(-\beta c x^2) (1 + \beta g x^3)$$

```
> int(f1(x),x=-infinity..infinity);
```

$$\begin{cases} \frac{3}{4} \frac{g\sqrt{\pi}}{\beta c^2 \sqrt{c\beta}} & \text{csgn}(c\beta) = 1 \\ \infty & \text{otherwise} \end{cases}$$

ここでは、 $c\beta$  が正の場合 ( $\text{csgn}(\beta c)=1$ ) とそれ以外の場合 (otherwise) に分けて答えを返している。しかしこのような意図したきれいな結果をいつも Maple が返してくれるとは限らない。これだけしか知らないと、なにかうまくいかないときにお手上げになってしまう。このようなきれいで簡単な結果に行き着く前の、裏でおこなういくつかの予備計算を省略せずに示そう。

先ず鉄則0にしたがって restart をかけ、関数を定義する。

```
> restart; f1:=unapply(x*exp(-beta*c*x^2)*(1+beta*g*x^3),x);
```

$$f1 := x \mapsto x \exp(-\beta c x^2) (1 + \beta g x^3)$$

次には鉄則 1 にしたがって積分する前にどのような関数かプロットしてみる。そのまま plot へ投げると怒られる。

```
> plot(f1(x),x=-10..10);
```

Warning, unable to evaluate the function to numeric values in the region; see the plotting command's help page to ensure the calling sequence is correct

これは鉄則 2 にあるとおり、数値を代入すれば、

```
> f1(10);
```


$$10 \exp(-100 c \beta) (1 + 1000 \beta g)$$

beta,c,g などのパラメータの値が入っていないためとわかる。適当に値を代入する。

```
> c:=1; g:=0.01; beta:=0.1; #res:1 0.01 0.1
```

再度プロットを試みる。

```
> plot(f1(x),x=-10..10);
```



実際に積分してみる。ここでは、鉄則 3 にしたがって、式を頭から打ち込むのではなく内側からみていくことが肝要である。これは問題を解いていく時に、思考が必ずたどるであろう順番に相当する。先ず変数に入れた数値をクリアする。

```
> c:='c': g:='g':beta:='beta':
```

不定積分でこの関数が積分できることを確認する。

```
> int(f1(x),x);
```

$$-\frac{1}{2} \frac{1}{\exp(c \beta x^2) \beta c} + \beta g \left( -\frac{1}{2} \frac{x^3 \exp(-c \beta x^2)}{c \beta} + \frac{3}{2} \left( -\frac{1}{2} \frac{x \exp(-c \beta x^2)}{c \beta} + \frac{1}{4} \frac{\sqrt{\pi} \operatorname{erf}(\sqrt{c \beta} x)}{c \beta \sqrt{c \beta}} \right) c^{-1} \beta^{-1} \right)$$

次に x=-alpha..alpha の定積分を実行する。これは上記のコマンドに付け足すようにしていく。

```
> int(f1(x),x=-alpha..alpha);
```

$$-\frac{1}{4} \frac{g (4 \alpha^3 \exp(-c \beta \alpha^2) \beta c \sqrt{c \beta} + 6 \alpha \exp(-c \beta \alpha^2) \sqrt{c \beta} - 3 \sqrt{\pi} \operatorname{erf}(\sqrt{c \beta} \alpha))}{\beta c^2 \sqrt{c \beta}}$$

さらに  $\alpha \mapsto \infty$  としてみる。

```
> limit(int(f1(x),x=-alpha..alpha),alpha=infinity);
```

$$\lim_{\alpha \rightarrow \infty} -\frac{1}{4} \frac{g \left( 4 \alpha^3 \exp(-c \beta \alpha^2) \beta c \sqrt{c \beta} + 6 \alpha \exp(-c \beta \alpha^2) \sqrt{c \beta} - 3 \sqrt{\pi} \operatorname{erf}(\sqrt{c \beta} \alpha) \right)}{\beta c^2 \sqrt{c \beta}}$$

ところがこれでは答えを返してくれない。積分した後のそれぞれの項を見ると  $\beta c > 0$  を仮定すれば簡単になることが分る。assume を使って、このような変数の仮定おこなう。

```
> assume(beta*c>0);
```

結果として最初に出した解答を得る。

```
> limit(int(f1(x),x=-alpha..alpha),alpha=infinity);
```

$$\frac{3}{4} \frac{\sqrt{\pi} g}{\beta c^2 \sqrt{\beta c}}$$

### 1.5.3 式のフォローのデフォルト

Maple で実際に数式をいじる状況というのは、ほとんどの場合が既知の数式変形のフォローだろう。例えば、論文で「(1) 式から (2) 式への変形は自明である」とかいう文章で済ましている変形が本当にあっているのかを確かめたい時、一番単純なやり方は自明と言われた前後の式が一致していることを確かめるだけで十分である。最も単純な確認法は以下の通り、変形の前後の式を手入力してその差を expand した結果が 0 か否かです。

```
> ex1:=(x-3)^4;
```

$$ex1 := (x - 3)^4$$

```
> ex2:=x^4-12*x^3+54*x^2-108*x+81;
```

$$ex2 := x^4 - 12x^3 + 54x^2 - 108x + 81$$

```
> expand(ex1-ex2); #res: 0
```

0 ならば式の変形は保証されているので、その導出が間違いでなく誤植などもないことが確認できる。ただ、これだけでは変形の哲学や技法が身に付くわけではない。あくまでも苦し紛れのデフォルトであることは心に留めておくように。論理値として確かめたいときには、evalb を使う。

```
> evalb(expand(ex1-ex2)=0); #res: true
```



## 1.6 センター試験 I(CenterExamI)

### [[[解説]]]

数式変形実践課題 (大学入試センター試験の解法を通して) 今まで出てきたコマンドを使えば, 典型的なセンター試験の問題を解くのも容易である. 以下の例題を参照して課題を解いてみよ. 使うコマンドは, unapply, solve, diff, expand(展開), factor(因数分解) と subs(一時的代入) である. expand 等の数式変形によく使うコマンドは次節以降で詳しく解説している. subs は以下を参考にせよ.

### 1.6.1 一時的代入 (subs)

代入 ( $:=$ ) が永続的なものに対して, 一時的な代入は subs で行う.

```
> subs(a=1,a+2); #res: 3
```

典型的な使い方は, solve で求めた解などを式 (equation) として代入しておいて, それを subs で一時的に当てはめる.

```
> eq1:=a=solve(a+b=0,a); subs(eq1,a+2);
```

$$a = -b$$

$$-b + 2$$

### [[[例題]]]

### 1.6.2 例題:2 次関数の頂点

$a, b$  を定数とし,  $a < 0$  とする. 2 次関数

$$y = ax^2 - bx - a + b \cdots (1)$$

のグラフが点  $(-2, 6)$  を通るとする. このとき

$$b = -a + \boxed{\text{ア}}$$

であり, グラフの頂点の座標を  $a$  を用いて表すと

$$\left( \frac{-a + \boxed{\text{イ}}}{\boxed{\text{ウ}}a}, -\frac{(\boxed{\text{エ}}a - \boxed{\text{オ}})^2}{\boxed{\text{カ}}a} \right)$$

である (2008 年度大学入試センター試験数学 I より抜粋).

### 解答例

まず, 与えられた 2 次関数を  $f(x)$  で定義する.

```
> restart; f:=unapply(a*x^2-b*x-a+b,x);
```

$$f := x \mapsto ax^2 - bx - a + b$$

与えられた点の座標を関数に入れる.

```
> eq1:=f(-2)=6;
```

$$eq1 := 3a + 3b = 6$$

これを b について解く.

```
> eq2:=b=solve(eq1,b);
```

$$eq2 := b = 2 - a$$

次は, 頂点の座標で傾きが 0 になることを用いて解いていく.

```
> solve(diff(f(x),x)=0,x);
```

$$\frac{1}{2} \frac{b}{a}$$

b の値は eq2 で求まっているので, それを代入 (subs) する.

```
> subs(eq2,solve(diff(f(x),x)=0,x));
```

$$\frac{1}{2} \frac{2-a}{a}$$

これを x0 として eq3 で定義しておく.

```
> eq3:=x0=subs(eq2,solve(diff(f(x),x)=0,x));
```

$$eq3 := x0 = \frac{1}{2} \frac{2-a}{a}$$

頂点の y 座標は,  $f(x0)$  で求まる

```
> f(x0);
```

$$ax0^2 - bx0 - a + b$$

eq2, eq3 で求まっている x0, b を代入する.

```
> eq4:=subs({eq2,eq3},f(x0));
```

$$eq4 := -\frac{1}{4} \frac{(2-a)^2}{a} - 2a + 2$$

これを因数分解 (factor) する.

```
> factor(subs({eq2,eq3},f(x0)));
```

$$-\frac{1}{4} \frac{(3a-2)^2}{a}$$

## [[[課題]]]

1.  $P = x(x+3)(2x-3)$  とする. また,  $a$  を定数とする.  $x = a+1$  のときの  $P$  の値は

$$2a^3 + \boxed{\text{ア}} a^2 + \boxed{\text{イ}} a - \boxed{\text{ウ}}$$

である.

$x = a+1$  のときの  $P$  の値と,  $x = a$  のときの  $P$  の値が等しいとする. このとき,  $a$  は

$$3a^2 + \boxed{\text{エ}} a - \boxed{\text{オ}} = 0$$

を満たす. したがって

$$a = \frac{\boxed{\text{カキ}} \pm \sqrt{\boxed{\text{クケ}}}}{\boxed{\text{コ}}}$$

である.

2. (例題1. に引き続いて,) さらに, 2次関数 (1) のグラフの頂点の  $y$  座標が  $-2$  であるとする. このとき,  $a$  は

$$\boxed{\text{キ}} a^2 - \boxed{\text{クケ}} a + \boxed{\text{コ}} = 0$$

を満たす. これより,  $a$  の値は

$$a = \boxed{\text{サ}}, \frac{\boxed{\text{シ}}}{\boxed{\text{ス}}}$$

である. 以下,  $a = \frac{\boxed{\text{シ}}}{\boxed{\text{ス}}}$  であるとする.

このとき, 2次関数 (1) のグラフの頂点の  $x$  座標は  $\boxed{\text{セ}}$  であり, (1) のグラフと  $x$  軸の2交点の  $x$  座標は  $\boxed{\text{ソ}}, \boxed{\text{タ}}$  である.

また, 関数 (1) は  $0 \leq x \leq 9$  において

$x = \boxed{\text{チ}}$  のとき, 最小値  $\boxed{\text{ツテ}}$  をとり,

$x = \boxed{\text{ト}}$  のとき, 最大値  $\frac{\boxed{\text{ナニ}}}{\boxed{\text{ヌ}}}$  をとる.

(2008 年度大学入試センター試験数学 I より抜粋).

## [[ 解答例 ]]

1.  $P$  を  $x$  の関数として定義,

> restart;

> P:=unapply(x\*(x+3)\*(2\*x-3),x);

$$P := x \mapsto x(x+3)(2x-3)$$

$P(a+1)$  および  $P(a)$  を形式的に出してみる.

> expand(P(a+1)), expand(P(a));

$$2a^3 + 9a^2 + 3a - 4, 2a^3 + 3a^2 - 9a$$

2 式を差し引く.

> eq1:=(expand(P(a+1))-expand(P(a)))/2;

$$eq1 := 3a^2 + 6a - 2$$

出題にそろえるため 2 で割っている. その式を eq1 として代入し,  $eq1=0$  を  $x$  について解く (solve).

> sol1:=solve(eq1=0,a);

$$sol1 := -1 + \frac{1}{3}\sqrt{15}, -1 - \frac{1}{3}\sqrt{15}$$

2. 例題の eq3,eq4 までを確認

> eq3, eq4;

$$x0 = \frac{1}{2} \frac{2-a}{a}, -\frac{1}{4} \frac{(2-a)^2}{a} - 2a + 2$$

eq4 が頂点の  $y$  座標の値なので, これから  $-2$  を引いて展開.

```
expand((eq4-(-2)));
```

$$-\frac{1}{a} + 5 - \frac{9}{4}a$$

これではわかりにくいので、出題にそう形にするため、 $-4a$  を掛け、eq5 とする.

```
> eq5:=expand((eq4+2)*(-4)*a);
```

$$eq5 := 4 - 20a + 9a^2$$

これを  $a$  について解いて (solve)

```
> solve(eq5=0,a);
```

$$2, \frac{2}{9}$$

$a = 2/9$  を eq3 に代入して、頂点の  $x$  座標を出す.

```
> subs(a=2/9,eq3);
```

$$x0 = 4$$

$a, b$  を  $f(x)$  に代入して,

```
> eq6:=subs({a=2/9,b=2-2/9},f(x));
```

$$eq6 := \frac{2}{9}x^2 - \frac{16}{9}x + \frac{14}{9}$$


これを解いて、 $x$  座標の交点を求める.

```
> solve(eq6=0,x);
```

$$7, 1$$

最大値, 最小値を求めるために、今まで求めたパラメータを代入して、plot してみる.

```
> plot(subs({a=2/9,b=2-2/9},f(x)),x=0..9);
```



目視で分かるとおり、最小値  $x = 4$ , 最大値  $x = 9$  で

```
> subs({a=2/9,b=2-2/9},f(4)), subs({a=2/9,b=2-2/9},f(9));
```

$$-2, \frac{32}{9}$$